

Criando menus personalizados no MS Access

Você alguma precisou criar um menu personalizado no Access? Se sim, como foi que você levou a cabo o processo?

Existem várias formas de se criar menus personalizados no Office e o MS Access nos fornece tabelas que podem ser utilizadas para armazenar os dados que serão utilizados para criar nossos menus.

Neste artigo, estarei mostrando como criar tais menus de forma eficiente utilizando uma conexão ao nosso banco de dados para carregar os itens que comporão o nosso menu.

Para iniciar, crie uma nova tabela, como mostra a figura abaixo:

	Field Name	Data Type	
🔑	mnuID	AutoNumber	
	TIPO	Number	Tipo de menu
	Application	Text	
	BeginGroup	Text	Propriedade para iniciar um novo grupo
	BuiltIn	Text	
	BuiltInFace	Text	
	Caption	Text	Texto de nosso botão, popup, etc
	Creator	Text	
	DescriptionText	Text	Descrição
	Enabled	Text	Define se o menu está habilitado ou não
	FaceID	Number	Face do botão
	OnAction	Text	Ação a ser executada quando o botão é clicado
	State	Number	
	Style	Text	
	Width	Number	

As propriedades principais estão comentadas para melhor compreensão. Esta tabela conterá os itens que comporão o nosso menu. Nomeie e salve a sua tabela.

Os dados agora podem ser entrados conforme a figura:

	mnuID	TIPO	BeginGroup	Caption	FaceID	OnAction	Width
	1	1		Popup 1			80
	2	2		Btn 1	326	acao	50
	3	2	TRUE	Btn 2	620	acao	50
	4	1		Popup 2			80
	5	2		Btn 3	986	acao	50
	6	2		Btn 4	234	acao	0
▶	Number)	0			0		0

Com os dados em nossa tabela, acrescente um módulo do VBA para entrarmos o código. Antes de iniciar, porém, você precisará instalar a referência da biblioteca do MS Office. No VBE do Access, vá até Ferramentas » Referências e instale o Microsoft Office X.x Object Library.

```

Sub mnu()
    Dim cmdBar As CommandBar
    Dim popup As CommandBarPopup
    Dim btn As CommandBarButton

    Dim cn As Object
    Dim rs As Recordset
    Dim Sql As String

    Set cn = Application.CurrentProject.Connection
    Sql = "SELECT * FROM [ItensMenu]"
    Set rs = CreateObject("ADODB.Recordset")
    rs.Open Sql, cn, 1

    On Error Resume Next
    CommandBars("MENU PRINCIPAL").Delete
    Set cmdBar = Application.CommandBars.Add _
        (Name:="MENU PRINCIPAL", Position:=msoBarFloating)

    If (rs.EOF) Then
        MsgBox "Não há itens para acrescentar ao menu!", vbInformation
        CommandBars("MENU PRINCIPAL").Delete
    Else
        While (Not (rs.EOF))
            Tipo = rs![Tipo]

            Select Case Tipo
                Case 1 'menu popup
                    Set popup = cmdBar.Controls.Add(Type:=msoControlPopup)
                    With popup
                        .Caption = rs![Caption]
                        .Width = rs![Width]
                    End With

                Case 2 'botão
                    Set btn = popup.Controls.Add(Type:=msoControlButton)
                    With btn
                        .BeginGroup = rs![BeginGroup]
                        .Caption = rs![Caption]
                        .FaceId = rs![FaceId]
                        .OnAction = rs![OnAction]
                        .State = rs![State]
                        .Width = rs![Width]
                    End With
            End Select
            rs.MoveNext

        Wend
    End If
    cmdBar.Visible = True
    cmdBar.Protection = msoBarNoCustomize + msoBarNoChangeDock + _
        msoBarNoHorizontalDock

    rs.Close
    Set rs = Nothing
    Set cn = Nothing
End Sub

```

O menu será construído baseado em cada caso encontrado no campo TIPO. Aqui, utilizo um valor numérico para definir o tipo do controle sendo inserido em nossa barra de comando. Contudo, o tipo pode ser uma string (texto) para melhor visualização. Por exemplo, poderíamos definir o caso para o botão como sendo BOTÃO, deixando o caso explicitamente definido. Lembre-se apenas que como estamos lidando com Strings o caso precisa estar entre aspas, isto é, "BOTÃO".

Se você deseja saber mais sobre criação de menus no MS Office, visite o site no pé desta página ou envie um email para autor.

Este exemplo faz parte de uma série de textos dedicados a construção de menus personalizados e integração no MS Office.